
第２章 

東アジア地域・貿易リンクモデル構築に伴うプログラム解説 

 

植村 仁一 

 

本章は、貿易リンクモデルの構築に伴うデータ処理・加工及び各国・地域の相手国

別、財別輸入関数の推定のために作成した エクセルファイル上で動作する Visual 

Basic for Application (VBA) プログラムの解説を行うものである。これは、今年度

を通じた作業に際して作成したプログラムの資料としての性格を兼ね、今後の貿易リ

ンクモデル作成の参考資料を提供するものである。一般に貿易リンクモデルの推定は、

国・財が複数あるために作業量が膨大なものになる傾向がある。そこで、これらプロ

グラムの整備をしておくことにより、モデル開発作業が効率的に進められると期待さ

れる。 

 

本章の構造は以下の通りである。まず、第１節で貿易リンクモデルの概要を解説し、

リンクモデル構築に必要なデータ群と、そこから推定される輸入関数群について、必

要とされる条件などを確認する。続く第２節では推定されるべき輸入関数群の共通構

造について概観する。第３節では輸入関数群の推定に必要である「競争者価格」の定

義と、実際の計算を行うためのプログラムの構造を解説する。第４～６節では、輸入

関数の満たすべき符号条件と、それを満たすための定式化（ここでは、変数変換やラ

グ構造の様々な組合せを指す）を模索するための省力化プログラムについて解説して

いる。 

 

第１節 貿易リンクモデルの概要 

 初めに、構築を目指す貿易リンクモデルを概観し、必要とされるデータとその定義

について解説する。 

 

 貿易リンクモデルとは、一つ一つが単体で完結している各国モデルを、各国間の貿

易構造を通じて接続することによって、全体を一つのマクロ計量モデルとみなすもの

である。 

 需要先決型の各国モデルでは、通常、「輸入価格」「輸出額」はモデル外から与えら

れる「外生変数」として扱われ、それを条件としてモデルを解き、均衡解としての「輸

出価格」「輸入額」が決定される（もちろん同時にその他の内生変数も決定される）。

ところで、複数の各国モデルを別々に取り扱う場合、ある国の輸出と別の国の輸入と

は直接的な関係はない。一方貿易リンクモデルは、各国モデルで決定された輸入額を

輸出国別に合算し、それを外生的に輸出国モデルに戻す、という手続きを踏むことに

より、各国の輸出入に有機的な関連を持たせようとするものである。 

 また、一つの国を輸入主体として見たとき、通常の場合その国が価格決定者である

ことは少ない。従って輸入価格は各国モデルでは外生的に扱われる。しかし、貿易リ

ンクの枠組みの中では、それは輸出国側の輸出価格（これは各国の事情である程度決

まると考える）の「加重平均」のような形で決まる。そこで貿易リンクモデルは、各

国の事情で決まる「輸出価格」「輸入額」を、相手国別に振り分けることにより、外生

的に各国に与えられる「輸入価格」「輸出額」を発生させるシステム、といってよい。 

 

 リンクシステムと、それを取り巻く各国モデルの関係は以下の概念図で表すことが

できる。この図では、中国モデルに注目した場合の各変数の流れを示している。なお、

章末附図１に概念図の詳細を提示してある。 

 

図１．リンクシステムの概念図 

 

 

 図中、実線は「財」の流れを、点線は「価格」の流れを示す。 

 

 現在構築を目指している貿易リンクモデルでは、輸入関数を相手国別・財別に推定

する。財の区分は一次産品：SITC 0,1,2,4、原油：SITC 3、製造業品：SITC 5-9 とす

るが、原油についてはすべて外生扱いとし、輸入関数の推定は行わない。また、関数

の定式化では、輸入の決定要因は、所得、輸入価格、輸出者の競争価格（第３節）と

する。 

 

第２節 輸入関数の定式化 

 上述の通り、貿易リンクモデルでは各国・地域の輸入関数を相手国別に推定してお

くことが必要となる。今回開発する貿易リンクモデルでは、説明変数として自国の所

得水準（GDP）、輸入価格と国内価格の相対価格（もしくは輸入価格の絶対水準）に加

え、相手国とその競争相手国群のオファー価格の相対価格を導入する。例えば一つの

財について、ある国の輸出価格が変化した場合、輸入者は、その国の輸出価格と、そ

の国の競争相手である国群の輸出価格とを比較し、より有利な条件で取引を行うとい

うことである。 

 国コード YYY の、XXX 国からの第 n財の輸入関数の定式化は、 

 

 YYYMnXXX = f[YYYGDP, YYYPMLCn/YYYPGDP, XXXPXn/XXXPCnYYY ] 

 

もしくは、輸入と国内の相対価格を用いない場合であれば、 

 

 YYYMnXXX = f[YYYGDP, YYYPMLCn, XXXPXn/XXXPCnYYY ] 

 

となる。ここで、変数名の頭につけられた国コードはモデル報告国のデータであるこ

とを示し、変数名の末尾の国コードは対象国を示す（以下の説明では、国を特定しな

い場合や明らかである場合は省略している）。YYYMnXXX は実質化された YYY 国の XXX

国からの輸入額（ドル建て）、YYYGDP は YYY 国の GDP（不変価格、現モデルでは 2000

年価格）、YYYPMLCn は第 n 財の輸入価格（現地通貨建て）、YYYPGDP は国内一般物価を

示し、同様に XXXPXn は XXX 国の輸出価格（対世界・ドル建て）である。XXXPCnYYY は

YYY 国市場において XXX 国が直面する、「XXX 国の競争国群の輸出価格」を表す。この

XXXPCnYYY は以下の式で定義される。 

 

XXXPCnYYY =  a ,YYY1 − aXXX,YYYXXX,YYY PX  

 

 ここで、a ,  は、第ｎ財のｊ国輸入市場における k国のシェアをを示す。従って、右

辺括弧内は、YYY 国を輸入国とする市場における XXX 国以外の参加者全体を 1 とした

時のシェア（ウェイト）を示すこととなる。、PX  は k 国の輸出価格である。 

 

なお、すべての k で世界全体をカバーしていれば、基準年におけるこの値は 1 とな

る。実際は世界全体をカバーしているわけではなく基準年の値は 1 と異なるため、運

用上はすべての値を基準年値で除して調整する。 

 

第３節 プログラム１：競争者価格の算出 

 上述の「一国市場におけるある輸出者に対する競争者価格」は、各国の対世界輸出

価格を、輸入国側の基準年の輸入額をウェイトとして加重平均を取ったものである。

定義式自体は非常に単純なものであるが、実際の計算手続きは煩雑である。そこで、

省力化のため、競争者価格を自動算出するプログラムを作成した。各国輸出価格に改

訂があった場合や入手可能時系列が伸びた場合、もしくは基準年の変更を行う場合な

ど、日々（年々）のルーチン的な作業においても省力化の点で有用である。 

 

 

３－１．ファイルの構造 

 エクセルファイルにはリンク参加国分の各国シート（15 枚）があり、ここには財別・

相手国別輸入額が格納されている。これとは別に各国の輸出価格が格納してあるシー

ト（2枚：一次産品と製造業品）と、輸入シェア行列が格納してあるシート（2枚）が

ある。輸入シェア行列シートでは、各国データシートから基準年に当たる輸入額を抽

出し、その年の輸入シェア行列を算出するようになっている。 

 

 さらに、競争者価格を格納するための空シートを 2 枚（PXC1 及び PXC3 シート）用

意しておく。また、ファイル全体の一枚目に、基準年入力と、参考用として計算の経

過時間を書き出すシートを用意しておく。 

 

３－２．計算作業と結果の出力 

 元データをすべて格納してある「完全版」は表中で多くの再計算が行われるため、

このファイル上でそのまま計算を行わせると速度が遅い。そこで、簡易版として国ご

とに必要部分を値複写した国別ファイルを用意する。具体的には輸入シェア行列シー

トに直接輸入シェア行列を入力しておくことで、15 枚の各国データシートがすべて不

要となる。実際の計算作業はそちらで行う（完全版でそのまま計算をした場合と比べ、

数 10 倍の速度となる）。 

一方完全版はデータの一元管理と 小限の計算を行うためのものとしておき、デー

タ改訂などがあった場合に、シェア行列などを自動計算するところまで用いる。 

 計算結果は PXC1 及び PXC3 シートに出力される。報告国と相手国が一致する場合、

競争者価格は当然存在しない。 

 

３－３．VBA スクリプト 

 章末附表１に全体を掲載している。全体の構造は以下のようになっている。 

 

(1) ComCode (1 or 3) による財の選択 

(2) 各国輸出価格読み込み 

(3) 輸入シェア行列の読み込み 

(4) 競争者価格の算出 

(5) 出力 

 

プログラムそのものは上から下へと流れるだけのものである。ComCode=1 で一次産品

を、ComCode=3 で製造業品を選択し、同一の手続きで計算が行われる。 

 

第４節 輸入関数の符号条件 

 各国の財別・相手国別輸入関数群は、符号条件を厳密に満たしておかないと、それ

らを組み合わせて貿易リンクモデルとする際、（所得が増えると輸入が減るというよう

な）経済学的におかしな挙動を取ってしまう懸念がある。実質輸入の推定式の係数に

期待される符号条件は、所得については正、輸入価格（もしくはそれと国内価格の比）

については負、輸出者と競争者のオファー価格比についても負である。すなわち、 

 

             (+)      (-)          (-) 

 MnXXX = f[ GDP, PMLCn/PGDP, XXXPXn/XXXPCn ] 

 

となる。これら輸入関数群はリンクモデルへの組み込みを主目的とするため、係数の

統計的有意性よりも符号条件を重要視する。 

 

 輸入関数の推定に当たっては、変数の変換を様々に組合せ、符号条件を満たすもの

を探していくという「総当たり」で臨み、条件を満たすものの中から 終的にモデル

に組み込む候補を絞り込んでいくというアプローチを行うこととする。変数変換は無

変換を含めて 4種類を想定する。 

 

(1) 無変換 

(2) 対数 

(3) 階差 

(4) 対数階差 

 

また、特に階差変換を行う (3) と (4) については、期待する符号が得られない場

合、説明変数として 1 期のラグ項を導入することにより、符号の逆転が期待される。

つまり、 

                          (+)          (+) 

 ΔMnXXX = f[ ・・・, ΔPMLCn, Δ(XXXPXn/XXXPCn) ] 

 

という（期待されない）符号となっている場合、それは右辺と左辺の今期の階差同士

が正の相関を持つことを意味する。ここで、説明変数を 1期のラグ項で置き換えれば、 

                            (-)               (-) 

 ΔMnXXX = f[ ・・・, (ΔPMLCn)(-1), (Δ(XXXPXn/XXXPCn))(-1) ] 

 

を得られやすくなる（必ずそうなるというわけではない）。 

 

第５節 輸入関数推定の省力化について 

 上述の変数変換とラグ項の導入を組み合わせると、ある国の、一つの財、一つの相

手国からの輸入関数一本の推定に関して、左辺・被説明変数と右辺・各説明変数につ

いて、 

（左辺） 

 無変換 対数 階差 対数階差 

 

（右辺第１項：所得） 

 無変換 対数 階差 対数階差 

 

（右辺第２項：輸入価格） 

 無変換 対数 階差 対数階差 無変換（１期ラグ） 対数（１期ラグ） 階差（１

期ラグ） 対数階差（１期ラグ） 

 

（右辺第３項：競争者とのオファー価格比） 

 無変換 対数 階差 対数階差 無変換（１期ラグ） 対数（１期ラグ） 階差（１

期ラグ） 対数階差（１期ラグ） 

 

だけの組合せが想定され、これだけで４×４×８×８＝１０２４通りとなる。これら

の中には、「水準」（無変換）の左辺を所得の対数で説明する、といったあまり意味を

持たない変数変換の組合せも存在している。実際のプログラム内ではそのような無意

味な定式化についてはスキップし、左辺と右辺第１項（所得）については同一の変数

変換をするという制約をつけている。 

 また、右辺第２項については輸入価格そのものか、国内との相対価格にするかとい

う選択があり、これも勘案すると（上記制約を無視すれば）２０４８通りの組合せと

なる。さらにそのそれぞれについて、被説明変数の１期ラグ項を説明変数として加え

るかどうかの選択を加えると４０９６通りもの定式化を比較することとなる。その中

から符号条件を満たし、かつ、統計的有意性のなるべく高いものを抽出するという作

業が必要となる。 

 

 このように、１本の式あたり４０９６通りの組合せについて符号条件を満たし、有

意性の高いものを選ぶところまでについてエクセルで自動的に行えるようにしたのが

今年度の一つの成果である。以下、その概要と一部詳細部分について解説し、記録に

残しておきたい。 

 

第６節 プログラム２：輸入関数推定省力化 

 エクセルには「回帰分析」を行う関数が装備されている。今回作成したプログラム

は、上記の被説明・説明変数の変数変換や回帰分析、符号条件の確認など、シート上

でできることはなるべくシート上の関数を用い、繰り返し計算や結果の書き出し等、

シート上では不可能な 低限と思われる部分だけを VBA スクリプトとして作成してい

る。章末附表２に全体を掲載する。 

 

６－１．ワークシートの構成 

 マクロ（VBA）付きエクセルファイルは、輸入側の国・地域（報告国）について一つ

ずつ用意する。シートの構成とそれぞれの主な機能は以下の通りである（番号は便宜

的につけてある）。 

 

(1) Data Center 

 報告国の基礎データ（GDP、GDP デフレーター、為替レート、財別ドル建て輸出入価

格）、及び実質化された輸入（財別、相手国別）を保存しておく。また、相手国のデー

タとしては輸出物価（財別）、相手国が報告国市場で直面する競争者価格（財別）も合

わせて保存しておく。このシートは米ドル建て輸出入価格から各国通貨ベースの輸出

入価格の算出を行うほか、(2) DepVar 以降のシートが必要に応じ参照している。 

 

(2) DepVar 

 被説明変数を抽出し、先に述べた変数変換を適用した系列を自動的に計算しておく。

回帰分析を行うシートからの信号で、適当な系列を返す。具体的には、「無変換」「対

数変換」「階差変換」「対数階差変換」の操作を行った列を順に並べておき、与えられ

た信号により DepVar シート内の別の列に必要な列を呼ぶ構造となっている。回帰分析

を行うシートからは常にこの列を呼ぶことにより、構造が繁雑となることを防いでい

る。 

 

(3) GDP 

(4) PM 

(5) Rel.PM 

(6) Rel.PXC 

 

これら４つについても同様である（１期ラグ項も同様）。(4)は絶対輸入価格、(5)

は相対輸入価格を作成しておくシートであり、この２つについては、回帰分析シート

からの信号により、どちらか片方の情報のみが返される。(6)は輸出国側の輸出価格と、

当該輸出国が報告国市場で直面する競争者の輸出価格との相対価格を計算するシート

である。 

 

(7) Data for OLS 

回帰分析を行うためのデータを一旦保存。(9) Regression シートにはそのまま読み

込まれるだけである。 

 

(8) Base 

 報告国を指定（ファイルごとに固定）。また、相手国（1～15）、財コード（1 or 3）、

相対輸入価格を使うか否か（使用=1、不使用=0）、説明変数の１期ラグを使うか否か（使

用=1、不使用=0）を選択するシート。回帰係数の t 値の目安の設定も行う。なお、実

際の運用に当たっては一々手作業で選択せず、報告国以外は VBA スクリプト内で自動

的に変化させて繰り返し計算を行う。 

 ところで、貿易データは相手国ごとに系列の開始年次が異なる場合があるため、回

帰分析を行うためのデータ系列の長さをすべての相手国で同一にしておくことができ

ない。このシートでは各相手国別のデータ開始年次を自動的に取得し、例外となる国

があるかどうかを調べることも行っている。例外となる長さを持つ相手国については、

次の (9) Regression シートにて分析期間を調整する（調整のみ手作業）。 

 

(9) Regression 

 必要なデータセットから回帰分析を行い、結果を評価する。具体的には回帰分析を

行うための LINEST(・) 関数を用いて係数と t値を計算し、「符号条件を満たしている

か」「設定した t値の Critical Value を超えているか」を判断する。変数変換と説明

変数のラグ項のすべての組合せ、すなわち、被説明変数の１期ラグ項を含めないケー

スで２０４８通り、含めるケースで２０４８通りの合計４０９６通りについて条件を

満たすものがない場合、t値の条件を 0.1 ずつゆるめて 初から計算をやり直す。t値

の条件は絶対値で 3.0 から始め、それが 0.0 となっても（符号条件の方で）一つも期

待される条件を満たすものが見つからない場合は計算を打ち切る。 

 期待される符号条件と、指定した大きさの t 値の Critical Value 以上であるか否

かの判定を行い、すべて満たされる場合に "Good!"、一つでも満たさなければ "NG" の

判定を下し、その結果は変数に格納しておき、後にその定式化を別シート（後述のシ

ート 10 から 13）に書き出す。これを、14 の相手国、財 1 と財 3、輸入価格の絶対価

格／相対価格別で次々に行っていく。 

 

（補足）有効なデータ系列の長さが異なる場合の処理について 

 相手国や財によって回帰分析に用いることのできるデータ系列の有効長が異なる場

合がある。この場合、すべてを「短い方」に合わせてしまうのはデータを無駄にする

ことになるし、自由度確保の点からもふさわしくない。 

 特に中国やベトナムについては貿易データが十分に長く取ることができないため、

この二国を相手国とする場合に有効なデータの長さが短くなる。全体で見ると、たい

ていの場合では「多くの国が同一の（長めの）データ長があり、一つか二つの（より

短い系列しか使えない）例外がある」ということが多い。そこで、このシートでは有

効長の異なる系列に対する回帰分析結果を複数準備しておき、相手国や財別といった

条件に従って参照先を変えるという方法を取る。 

 

 例として被説明変数のラグ項を説明変数に含まず、韓国と台湾、ベトナムではその

他多くの国よりも短い（しかもベトナムの第 3 財のみはさらに短い）系列しか使うこ

とのできない中国について、その概要を記録しておく。 

 

 この例では、3 カ所にそれぞれデータ長の異なる回帰分析を行った結果を置いてい

る。そして、横に立てられたフラグにより、 終的な計算結果として使うものを取り

出す構造としている。すなわち、 

（イ）Majority に関する係数及び t値（1985-2009：分析結果） 

（ロ）ベトナムの第 1財及び韓国、台湾に関する係数及び t値（1987-2009：分析結果） 

（ハ）ベトナムの第 3財に関する係数及び t値（1988-2009：分析結果） 

 

は常に算出されている。従ってこの場合、1985 年からのデータを使うように指定され

てある（イ）の欄は、（ロ）や（ハ）が 終結果として採用されるようなケース（1985

年や 1986 年には N/A 値が入っている）では回帰係数は算出されず、エラーとなって

いるが、そこをフラグで回避して拾わないようにしている。具体的には、セル F29 に

は「国コード＝VNM かつ財コード＝1」または「国コード＝KOR」または「国コード＝

TWN」の３通りの時にフラグ 1が立ち、セル F35 には「国コード＝VNM かつ財コード＝

1」の場合のみフラグ 1が立つ。セル F23 は 1-(F29+F35) という式が入力されており、

上記の場合以外すべてについてフラグ 1が立つ。 

 LINEST（・）関数の出力先はこれに応じて３通り×２（被説明変数の１期ラグ項を

説明変数として導入するか否か）の６通り準備されており、それぞれ、上記３×２＝

６通りの期間（ラグ項入りの場合は１年分短くなる）について計算範囲を選択してお

く。この部分のみが手作業で行うべき操作であり、計算と出力は自動で行われる。 

 

(10) Out(AO) 

 条件を満たす定式化が見つかった場合の出力用シート。以下も同じ。 

 

(11) Out(AW) 

(12) Out(RO) 

(13) Out(RW) 

 

 それぞれ、絶対価格・被説明変数のラグ項なし（Absolute/withOut）、相対価格・被

説明変数のラグ項あり（Relative/With）などという意味である。 

 

(14) Summary 

 相手国別、財別、絶対／相対価格別、被説明変数のラグ項の有無の別で、t 値がい

くらの時に初めて条件を満たすものがいくつ見つかったかを書き出しておくシート。

同時にそれぞれの相手国別でかかった時間も計算するようにしてある。 

 

６－２．VBA スクリプト 

 章末附表２に全体を掲載しているが、ワークシート上では実行不可能であるか、可

能であるとしても非効率である部分を VBA で自動処理するようにしている。全体の構

造は以下のようになっている。繰り返し計算のループは５重である 

 

(1) 変数宣言部分 

(2) 報告国の設定 

(3) ループ 

   (a) 相手国 

   (b) 輸入財 

   (c) 被説明変数のラグ項導入有無 

   (d) 絶対/相対輸入価格切り替え 

   (e) t 値 critical value の変更 

(4) 条件を満たす定式化候補の書き出し 

(5) サマリーシートへの情報書き出し 

 

初めに、T_Critical として判定に用いるための t 値の Critical Value を設定す

る（初期値は 3.0）。条件を満たすものがなければ、この値を 0.1 ずつ引き下げて計算

を行う。 初から「候補」が多すぎる場合はこの値を 6.0 などに上げても良いが、反

対に、ヒットするはずのものが低い t 値から始まる場合には時間がそれだけ浪費され

ることとなる。 

 

また、刻み幅が 0.1 では多すぎる候補が出てしまう場合（例えば T_Critical=1.3 で

は候補は一つもなかったが、T_Critical=1.2 では数十の候補が出た、など）には、

T_Critical を小数第二位まで設定し、再計算することも可能にしてある。上の例の場

合では、例えば T_Critical=1.29 として計算を始めると、刻み幅が 0.01 となり、

T_Critical=1.2 となったところで計算を終わるようになる。実際にはそれより前のど

こかで、より少ない候補を与える t値が決まることを想定している。 

 

第７節 輸入関数推定の実際（中国の各相手国からの第 1財輸入） 

 実際に、中国のデータを適用して輸入関数の推定を行った結果は次の通りである。

財と相手国の組合せ、被説明変数のラグ項導入の有無ごとに、それぞれ４０９６通り

の定式化で推定し、係数の有意性を t 値で判定している。t 値が１以上となったもの

のみ具体的な定式化を示す。ここでは中国の第 1 財輸入について例示する。全体の結

果はデータ篇４にまとめて掲載している。表題の右にある数値は符号条件を満たした

定式化全体の数を、左端の○、×はラグ項の有無を示し、その次の数字は条件を満た

した場合の 大 t値である。 

 ここでは右辺第２項を「輸入価格／国内価格」の相対価格で推定している。表には

示していないが、この部分を輸入価格単体（絶対価格）でも推定するため、実際の計

算量はさらに倍となる。 

 

（例）第 1財・対日本（×64、○46） 

（ラグ項なしで 64 本、ラグ項ありで 46本の式が符号条件を満たしている） 

  (×1.7) M1JPN =f[ GDP, Δlog(PM1LC(-1)/PGDP(-1)), Δlog(JPNPX1/JPNPC1) ] 
  (○1.2) M1JPN =f[ GDP, Δlog(PM1LC(-1)/PGDP(-1)), Δlog(JPNPX1/JPNPC1), 

 M1JPN(-1) ] 

（例）第 1財・対韓国（×66、○62） 

  (×1.7) log(M1KOR) =f[ log(GDP), PM1LC/PGDP, Δ(KORPX1/KORPC1) ] 
  (×1.7) Δ(M1KOR) =f[ Δ(GDP), Δlog(PM1LC/PGDP), Δ(KORPX1/KORPC1) ] 
  (○2.2) M1KOR =f[ GDP, log(PM1LC/PGDP), Δ(KORPX1/KORPC1), M1KOR(-1) ] 
  (○2.2) M1KOR =f[ GDP, log(PM1LC/PGDP), Δlog(KORPX1/KORPC1), M1KOR(-1) ] 
 
（例）第 1財・対台湾（×66、○16） 

  (×3.0) M1TWN =f[ GDP, PM1LC/PGDP, TWNPX1/TWNPC1 ] 
  (×3.0) M1TWN =f[ GDP, log(PM1LC/PGDP), TWNPX1/TWNPC1 ] 
  (×3.0) M1TWN =f[ GDP, log(PM1LC/PGDP), log(TWNPX1/TWNPC1) ] 
  (○0.7) log(M1TWN) =f[ log(GDP), PM1LC/PGDP, TWNPX1/TWNPC1, 

 log(M1TWN(-1)) ] 
  (○0.7) log(M1TWN) =f[ log(GDP), PM1LC/PGDP, log(TWNPX1/TWNPC1), 

 log(M1TWN(-1)) ] 
 
（例）第 1財・対米国（×79、○58） 

  (×2.0) M1USA =f[ GDP, Δlog(PM1LC/PGDP), USAPX1/USAPC1 ] 
  (×2.0) M1USA =f[ GDP, Δlog(PM1LC/PGDP), log(USAPX1/USAPC1) ] 
  (○1.1) M1USA =f[ GDP, Δ(PM1LC/PGDP), USAPX1/USAPC1, M1USA(-1) ] 
  (○1.1) M1USA =f[ GDP, Δ(PM1LC/PGDP), log(USAPX1/USAPC1), M1USA(-1) ] 
  (○1.1) M1USA =f[ GDP, Δ(PM1LC/PGDP), Δ(USAPX1/USAPC1), M1USA(-1) ] 
  (○1.1) M1USA =f[ GDP, Δ(PM1LC/PGDP), Δlog(USAPX1/USAPC1), M1USA(-1) ] 
 
 このように、ある程度機械的に輸入関数の「候補」を絞り込んでおけば、それらを

実際にモデルに組み込み、モデル全体のパフォーマンスの優劣や安定性などを相互に

比較することによって、最終的にモデルに用いる輸入関数を決定していく作業が容易

になる。 

 

第８節 輸入関数推定・高速版の開発に向けて 

 上記プログラムは、各係数について符号条件を満たし、t 値がより高いものを抽出

するために、t 値を徐々に変化（低下）させつつすべての条件を満たすものを抽出し

ていた。高速化を目指した次バージョンでは、t 値を 3.0 から 0.1 刻みに 0.1 までの

30 段用意し、一つの定式化に対して 30 段分の係数算出と判定を並列的に行い、条件

を満たすもののうち最大の t 値を取るものを抽出する。t 値のループが１段減ること

になる分高速化されることが期待される。 

【附図１】 リンクシステムの詳細図 

【附表１】競争者価格の算出プログラム（VBA スクリプト） 

 

 

Sub MakePXC() 
 
' ---------- Magic Numbers ---------- 
' 定数設定 
' 
' データ開始年、終了年および基準年の設定 
Const YearBegin = 1970, YearEnd = 2009, BaseYear = 2000 
' 
' リンク参加国数 
Const NCntry = 15 ' <<<<<=====  Number of total countries 
 
' --------------------------------- 
' 時系列の長さ（年） 
Dim Nyear As Byte 
Nyear = YearEnd - YearBegin + 1 
 
' --------------------------------- 
' 参加国名コード 
Dim CountryName(NCntry) As String * 3 
 
'Share, PX シート内で↓この順に国が並んでいる必要あり 
CountryName(1) = "AUS" 
CountryName(2) = "CHN" 
CountryName(3) = "HKG" 
CountryName(4) = "IDN" 
CountryName(5) = "JPN" 
CountryName(6) = "KOR" 
CountryName(7) = "MYS" 
CountryName(8) = "NZL" 
CountryName(9) = "PHL" 
CountryName(10) = "SGP" 
CountryName(11) = "THA" 
CountryName(12) = "TWN" 
CountryName(13) = "USA" 
CountryName(14) = "VNM" 
CountryName(15) = "EUR" 

' --------------------------------------- 
' 輸入シェア行列の読み込み 
'     << Read Share Matrix >> 
' --------------------------------------- 
' 一次産品（Share1）、工業品（Share2） 
Dim ComCode As Byte 
Dim Share(NCntry, NCntry, 2) As Double 
 
Sheets("基準年入力").Select 
Range("a7").Select 
Selection.Value = "Data Read" 
Range("a8").Select 
Selection.Value = "   Begin: " & Now 
 
For ComCode = 1 To 2 ' ComCode=1: Primary, ComCode=3: Manufacturing 
 
Select Case ComCode 
    Case 1 
        Sheets("Share 0124").Select 
    Case 2 
        Sheets("Share 5-9").Select 
End Select 
 
    For zi = 1 To NCntry - 1            ' R.cntry は EUR なし（別に定義） 
        For zj = 1 To NCntry 
            Range("a2").Select 
            ActiveCell.Offset(zi, zj).Select 
            Share(zi, zj, ComCode) = Selection.Value 
        Next zj 
    Next zi 
 
' --------------------------------- 
'     << Read Export Price >> 
' --------------------------------- 
Dim PX(YearEnd - YearBegin + 1, NCntry, 2) As Double 
 
Dim BasePoint As Byte 
Dim BasePrice As Double 
 
BasePoint = BaseYear - YearBegin + 1 

Select Case ComCode 
    Case 1 
        Sheets("PX1").Select 
    Case 2 
        Sheets("PX3").Select 
End Select 
 
For zi = 1 To Nyear 
    For zj = 1 To NCntry 
        Range("a1").Select 
        ActiveCell.Offset(zi, zj).Select 
        PX(zi, zj, ComCode) = Selection.Value 
    Next zj 
Next zi 
 
For zj = 1 To NCntry 
    BasePrice = PX(BasePoint, zj, ComCode) 
    For zi = 1 To Nyear 
        PX(zi, zj, ComCode) = PX(zi, zj, ComCode) / BasePrice 
    Next zi 
Next zj 
 
Sheets("基準年入力").Select 
Range("a9").Select 
Selection.Value = "    End:  " & Now 
 
' --------------------------------- 
'        << Main Routine >> 
' --------------------------------- 
Dim PXC(NCntry, YearEnd - YearBegin + 1, NCntry, 2) As Double 
 
Dim SMtmp(NCntry, 2) As Double 
Dim Denom(2) As Double 
Dim TmpVal(2) As Double 
 
Sheets("基準年入力").Select 
Range("a10").Select 
Selection.Value = "Calc." 
Range("a11").Select 

Selection.Value = "   Begin: " & Now 
 
For i = 1 To NCntry                       ' in the ith market 
    For j = 1 To NCntry                   ' reporting country 
        For zj = 1 To NCntry              ' import share of PC 
            SMtmp(zj, 1) = Share(i, zj, 1) 
            SMtmp(zj, 2) = Share(i, zj, 2) 
        Next zj 
        Denom(1) = 1 - SMtmp(j, 1) 
        Denom(2) = 1 - SMtmp(j, 2) 
        ' EU については「自国」も「競争者」に入れる 
        If j = 15 Then                      ' except for EUR as RC 
        ' すなわち、EU 以外について「自国」のシェアを 0 とする 
            Else 
                SMtmp(j, 1) = 0              ' force to be 0 for RC 
                SMtmp(j, 2) = 0 
        End If 
 
        For t = 1 To Nyear 
            TmpVal(1) = 0:   TmpVal(2) = 0 
            For zj = 1 To NCntry 
                For zk = 1 To 2 
                TmpVal(zk) = TmpVal(zk) + (SMtmp(zj, zk) /Denom(zk)) * PX(t, zj, zk) 
                Next zk 
            Next zj 
            If j = i Then 
                    PXC(i, t, j, 1) = 0 
                    PXC(i, t, j, 2) = 0 
                Else 
                    PXC(i, t, j, 1) = TmpVal(1) 
                    PXC(i, t, j, 2) = TmpVal(2) 
            End If 
        Next t 
    Next j 
Next i 
 
Sheets("基準年入力").Select 
Range("a12").Select 
Selection.Value = "    End:  " & Now 

' ---------------------------------- 
'      << Output >> 
' ---------------------------------- 
Sheets("基準年入力").Select 
Range("a13").Select 
Selection.Value = "Output" 
 
Select Case ComCode 
    Case 1 
        Range("a14").Select 
        Selection.Value = "   PXC1 Begin: " & Now 
    Case 2 
        Range("a16").Select 
        Selection.Value = "   PXC3 Begin: " & Now 
End Select 
 
' 財別による出力先の変更 
Select Case ComCode 
    Case 1 
        Sheets("PXC1").Select 
    Case 2 
        Sheets("PXC3").Select 
End Select 
 
GoSub HideAreas 
 
For i = 1 To NCntry          ' in the ith market 
    Range("a1").Select 
    ActiveCell.Offset((i - 1) * (Nyear + 1), 0).Select 
Selection.Value = CountryName(i) 
    For j = 1 To NCntry      ' reporting cntry 
        Range("a1").Select 
        ActiveCell.Offset((i - 1) * (Nyear + 1), j). Select 
Selection.Value = CountryName(j) 
        For t = 1 To Nyear 
            Range("a1").Select 
            ActiveCell.Offset((i - 1) * (Nyear + 1) + t, 0).Select 
            Selection.Value = YearBegin + (t - 1) 
            Range("a1").Select 
            ActiveCell.Offset((i - 1) * (Nyear + 1) + t, j).Select 

' --------------------------------------------------------------------------- 
            Selection.Value = PXC(i, t, j, ComCode) 
' --------------------------------------------------------------------------- 
         
        Next t 
    Next j 
Next i 
 
GoSub RevealAreas 
 
Sheets("基準年入力").Select 
Select Case ComCode 
    Case 1 
        Range("a15").Select 
        Selection.Value = "   PXC1  End:  " & Now 
    Case 2 
        Range("a17").Select 
        Selection.Value = "   PXC3  End:  " & Now 
End Select 
 
GoTo SkipSubRoutine  ' --- subroutine section --- 
' -------------------------------------- 
'    Subroutine Storage 
' -------------------------------------- 
HideAreas: 
        Rows("2:1000").EntireRow.Hidden = True 
        Columns("b:p").EntireColumn.Hidden = True 
    Return 
 
RevealAreas: 
        Rows("2:1000").EntireRow.Hidden = False 
        Columns("b:p").EntireColumn.Hidden = False 
    Return 
SkipSubRoutine: ' --- end of subroutines --- 
Next ComCode 
 
Sheets("基準年入力").Select 
Range("a5").Select 
 
End Sub 

【附表２】輸入関数推定の省力化（VBA スクリプト） 
 
 Excel VBA (Visual Basic for Application) は行番号によって管理する必要のない構造

化言語であるが、ここに限り説明の便宜のために左端に行番号を附加している。 
 
00010 Sub StartReg() 
00020 ' 国名：Reporting Country/ Partner Countries 
00030 Dim RCountry As String 
00040 Dim PCntryName(15) As String 
00050  
00060 ' 回帰分析を行う際の変数変換の条件格納変数 
00070 ' それぞれ、Regression シートにおけるセル名としてある 
00080 Dim H2, J2, K2, L2 As Byte 
00090 Dim K3, L3 As Byte 
00100  
00110 ' 判定結果格納用。1 ループにつき 1024 通りなので配列は 2000 準備 
00120 Dim Judgement(2000) As String 
00130  
00140 ' 定式化の変数名格納用 
00150 Dim DepVar(2000) As String 
00160 Dim IndepVar(2000, 4) As String 
00170  
00180 ' 諸条件をまとめて格納（一時的な保存用） 
00190 Dim Flags(2000) As String 
00200  
00210 ' 条件を満たした（良い）結果の番号保存 
00220 Dim GoodSpec(2000) As Boolean 
00230  
00240 ' カウンター変数 
00250 Dim Counter As Integer 
00260  
00270 ' 条件を満たした（良い）結果の数を数える 
00280 Dim GoodCounter As Integer 
00290  
00300 ' 結果書き出しに際し、元のセル（列）への復帰用カウンター 
00310 Dim ColSft As Byte 
00320  
00330 ' タイマー 
00340 Dim BeginTime, Endtime As Variant 
00350  

00360 Dim SingleSpecFlag As Boolean 
00370  
00380 ' 満たすべき t 値の設定・再設定用 
00390 Dim T_Critical, T_Increment As Double 
00400 Dim T_CritForce As Double 
00410  
00420 ' t 値の初期設定値と処理終了用ストッパー値用 
00430 Dim T_Init, T_Term As Double 
00440  
00450 ' 財コード、絶対/相対輸入価格、被説明変数ラグ項導入の有無（2 値） 
00460 Dim ComCode, RelativePM, WithWithoutLag As Byte 
00470 Dim PCntryForce As String 
00480  
00490 ' 一国のみ再計算する場合の再設定用 
00500 Dim ComCodeBegin, RelativePMBegin, WithWithoutLagBegin As Byte 
00510 Dim ComCodeEnd, RelativePMEnd, WithWithoutLagEnd As Byte 
00520 Dim PCntryBegin, PCntryEnd As Byte 
00530  
00540 Dim PCntrySelect(15) As Byte 
00550 Dim zi As Byte 
00560  
00570 ' 結果を書き出すシート選択用スイッチ（4 種類の値を取る） 
00580 Dim RelPMJump As Byte 
00590  
00600 ' 国番号保存用 
00610 Dim QAUS, QCHN, QHKG, QIDN, QJPN, QKOR, QMYS, QNZL As Byte 
00620 Dim QPHL, QSGP, QTHA, QTWN, QUSA, QVNM, QEUR As Byte 
00630  
00640 ' ループのカウンター及び一時格納用ダミー 
00650 Dim i, ii As Integer 
00660 Dim xxx As String ' dummy 
00670  
00680 ' 変数の初期化 
00690 ' ----------------------- init ---------------------- 
00700  QAUS = 1: QCHN = 2: QHKG = 3: QIDN = 4: QJPN = 5: QKOR = 6 
00710  QMYS = 7: QNZL = 8: QPHL = 9: QSGP = 10: QTHA = 11: QTWN = 12 
00720  QUSA = 13: QVNM = 14: QEUR = 15 
00730   
00740 ' 変数名（国名）の設定 
00750  PCntryName(QAUS) = "AUS": PCntryName(QCHN) = "CHN" 

00760  PCntryName(QHKG) = "HKG": PCntryName(QIDN) = "IDN" 
00770  PCntryName(QJPN) = "JPN": PCntryName(QKOR) = "KOR" 
00780  PCntryName(QMYS) = "MYS": PCntryName(QNZL) = "NZL" 
00790  PCntryName(QPHL) = "PHL": PCntryName(QSGP) = "SGP" 
00800  PCntryName(QTHA) = "THA": PCntryName(QTWN) = "TWN" 
00810  PCntryName(QUSA) = "USA": PCntryName(QVNM) = "VNM" 
00820  PCntryName(QEUR) = "EUR" 
00830  
00840 ' 報告国（自国）のワークシート上への設定 
00850 Sheets("Base").Select 
00860 RCountry = Range("a1").Value 
00870  
00880 ' Base シートにて一国のみの再計算を指定する場合 
00890 PCntryForce = Range("l3").Value 
00900 ComCodeBegin = Range("m3").Value 
00910 RelativePMBegin = Range("n3").Value 
00920 WithWithoutLagBegin = Range("o3").Value 
00930 T_CritForce = Range("L6").Value 
00940  
00950 T_Init = 3#  ' -------- initial value (T-Critical Value) --------- t 値の初期値 
00960 SingleSpecFlag = False 
00970 If PCntryForce = "" Then          ' *** 「1 国のみ」の条件がない場合 
00980         PCntryBegin = 1 
00990         PCntryEnd = 15 
01000         ComCodeBegin = 1 
01010         RelativePMBegin = 0 
01020         WithWithoutLagBegin = 0 
01030         ComCodeEnd = 3 
01040         RelativePMEnd = 1 
01050         WithWithoutLagEnd = 1 
01060         T_Increment = 0.1 
01070         T_Term = 0 
01080     Else                                ' *** 「1 国のみ」の条件がある場合 
01090         SingleSpecFlag = True 
01100         For i = 1 To 15 
01110             If PCntryForce = PCntryName(i) Then 
01120                 PCntryBegin = i 
01130             Else 
01140             End If 
01150         Next i 

01160         PCntryEnd = PCntryBegin 
01170         ComCodeEnd = ComCodeBegin 
01180         RelativePMEnd = RelativePMBegin 
01190         WithWithoutLagEnd = WithWithoutLagBegin 
01200         T_Increment = 0.1 
01210         T_Init = T_CritForce 
01220 End If 
01230 T_Critical = T_Init 
01240  
01250  
01260 '************************************************ 
01270 ' 特定の国だけを対象とする場合の設定 
01280 ' この場合は、下の zi ループで回し、ii ループを 
01290  コメントアウト（1340 行と 4130 行） 
01300  （全体で行う場合は 1320 行と 4140 行をコメントアウト） 
01310 '************************************************ 
01320 PCntrySelect(1) = QCHN: PCntrySelect(2) = QJPN 
01330 PCntrySelect(3) = QKOR: PCntrySelect(4) = QTWN 
01340 PCntrySelect(5) = QUSA 
01350  
01360 For zi = 2 To 3 
01370 ii = PCntrySelect(zi)  
01380 ' For ii = PCntryBegin To PCntryEnd  ' （コメントアウト） 
01390 ' 相手国別輸入関数の推定時、全部で 15 の国・地域があるが、 
01400 ' その中で「自分自身」が相手国となった場合にはスキップ（4120 行へ） 
01410 If PCntryName(ii) = RCountry Then GoTo SkipThisCountry 
01420  
01430 Sheets("Base").Select 
01440 Range("b3").Select 
01450 Selection.Value = PCntryName(ii) 
01460 PCntry = Selection.Value 
01470  
01480 Range("b7").Value = T_Critical 
01490  
01500 ' 計算開始時刻設定 
01510 BeginTime = Time 
01520 '************************************************************** 
01530 ' ----------- ComCode Loop       （ループ 2/5） ------------ 
01540 ' 財コード「１」「３」を発生 
01550 For ComCode = ComCodeBegin To ComCodeEnd Step 2 

01560 Sheets("Base").Select 
01570 Range("d3").Select 
01580 Selection.Value = ComCode 
01590  
01600 ' --------- With/Wighout Lag Loop （ループ 3/5） ---------- 
01610 ' 被説明変数のラグ項を説明変数に含めるか 
01620 For WithWithoutLag = WithWithoutLagBegin To WithWithoutLagEnd 
01630 Sheets("Base").Select 
01640 Range("h3").Select 
01650 Selection.Value = WithWithoutLag 
01660  
01670 ' --------- Relative PM Loop        （ループ 4/5） ---------- 
01680 ' 絶対/相対輸入価格を用いるか 
01690 For RelativePM = 1 To 1 '              RelativePMBegin To RelativePMEnd 
01700 Sheets("Base").Select 
01710 Range("f3").Select 
01720 Selection.Value = RelativePM 
01730 ' 
01740  
01750 T_Critical = T_Init ' refresh t-critical value 
01760 ' -------------------- increment change ---------------------- 
01770 ' 判定用 t 値の初期設定値（通常開始時は 3.0） 
01780 ' 初期設定値が小数第二位まで設定されていた場合、そこから 
01790 ' 刻み幅 0.01 で切りの良い数字まで下げる。例えば T_Critical = 1.39 と 
01800 ' 設定すると、T_Term = 1.3 までで計算を打ち切る。 
01810 ' 
01820 Sheets("Base").Select 
01830 If T_Critical * 100 - Int(T_Critical * 10) * 10 <> 0 Then 
01840         T_Increment = 0.01 
01850         T_Term = Int(T_Critical * 10) / 10 - 0.01 ' --- Stopper --- 
01860     Else 
01870         T_Increment = 0.1 
01880         T_Term = 0 
01890 End If 
01900  
01910 Sheets("Regression").Select 
01920  
01930 Range("d2").Select 
01940 Selection.Value = T_Term 
01950  

01960 GoodCounter = 0 
01970 ' ループ 5/5 
01980 ' ------------------------------------------------------ 
01990 ' ------ メインルーチン（ここから）-------- 
02000 ' ------------------------------------------------------ 
02010 Do While GoodCounter = 0 ' ------- regression loop ------- 
02020 Sheets("Base").Select 
02030 Range("b7").Select 
02040 Selection.Value = T_Critical 
02050  
02060 Sheets("Regression").Select 
02070  
02080 Counter = 0 
02090 GoodCounter = 0 
02100 ' -------------------- 変数変換 -------------------- 
02110 ' --- 被説明変数 --- 
02120 For H2 = 0 To 3                               ' サブループ(5) 1/6 
02130     Range("h2").Select 
02140     Selection.Value = H2 ' DepVar Transformation 
02150     For J2 = 0 To 3                           ' サブループ(5) 2/6 
02160     ' GDP は被説明変数と同じ変数変換とする。それ以外はスキップ（2600 行へ） 
02170     If J2 <> H2 Then GoTo SkipThisTransformation 
02180         Range("j2").Select 
02190         Selection.Value = J2 ' Income 
02200         For K2 = 0 To 3                       ' サブループ(5) 3/6 
02210             Range("k2").Select 
02220             Selection.Value = K2 ' PM 
02230             For L2 = 0 To 3                   ' サブループ(5) 4/6 
02240                 Range("l2").Select 
02250                 Selection.Value = L2 ' PX/PXC 
02260                 ' --------------- 1 期ラグ項の使用 ----------------- 
02270                 ' --- 輸入価格 --- 
02280                 For K3 = 0 To 1               ' サブループ(5) 5/6 
02290                     Range("k3").Select 
02300                     Selection.Value = K3 
02310                     ' --- オファー価格と競争者価格比 --- 
02320                     For L3 = 0 To 1           ' サブループ(5) 6/6 
02330                         Range("l3").Select 
02340                         Selection.Value = L3 
02350                          

02360                         Counter = Counter + 1 
02370                         GoodSpec(Counter) = False 
02380                         Judgement(Counter) = Range("e3").Value 
02390                             If Range("e3") = "Good!" Then 
02400                                 GoodSpec(Counter) = True 
02410                                 GoodCounter = GoodCounter + 1 
02420                                 DepVar(Counter) = Range("h4").Value 
02430                                 IndepVar(Counter, 1) = Range("j4").Value 
02440                                 IndepVar(Counter, 2) = Range("k4").Value 
02450                                 IndepVar(Counter, 3) = Range("l4").Value 
02460                                 IndepVar(Counter, 4) = Range("m4").Value 
02470                                 Else 
02480                             End If 
02490                             Range("f3").Value = GoodCounter 
02500                             ' 諸条件をまとめて格納 
02510                             Flags(Counter) = H2 
02520                             Flags(Counter) = Flags(Counter) & " " & J2 
02530                             Flags(Counter) = Flags(Counter) & " " & K2 
02540                             Flags(Counter) = Flags(Counter) & " " & L2 
02550                     Flags(Counter) = Flags(Counter) & " / " & K3 & " " & L3 
02560                     Next L3            ' (5) 6/6 終わり 
02570                 Next K3                ' (5) 5/6 終わり 
02580             Next L2                    ' (5) 4/6 終わり 
02590         Next K2                        ' (5) 3/6 終わり 
02600 SkipThisTransformation: ' GDP が被説明変数と違う変換の場合のスキップ先 
02610     Next J2                            ' (5) 2/6 終わり 
02620 Next H2                                ' (5) 1/6 終わり 
02630  
02640 ' ------------------- T-Critical Value Reset --------------------- 
02650 ' 条件を満たすものがなければ t 値の Critical Value を決められた 
02660 ' 刻みで引き下げ、新たなループに 
02670 T_Critical = T_Critical - T_Increment 
02680  
02690 ' その過程で Critical Value が設定した終了値となったら（符号条件を 
02700 ' 満たすものが一つもなければ）この関数の推計自体をスキップ 
02710 If T_Critical < T_Term Then GoTo SkipThisSpec  
02720  
02730 Loop  ' ------------ regression loop ------------------------ 
02740 ' -------------------------------------------------------------------- 
02750 ' ------------ メインルーチン（ここまで）--------------- 

02760 ' ------------------------------------------------------------------- 
02770 ' ------------ 一つの定式化に対する推定（ループ 5/5：終端) ------------ 
02780 SkipThisSpec:  ' === 符号条件を満たす定式化がない場合のスキップ先 === 
02790  
02800 ' -------------------------------- output --------------------------------- 
02810 ' 絶対/相対輸入価格及び被説明変数のラグ項の有無による 
02820 ' スイッチ変数の設定（0～3） 
02830 RelPMJump = 2 * RelativePM + WithWithoutLag 
02840 Select Case RelPMJump 
02850         Case 0 
02860             Sheets("Out(AO)").Select 
02870         Case 1 
02880             Sheets("Out(AW)").Select 
02890         Case 2 
02900             Sheets("Out(RO)").Select 
02910         Case 3 
02920             Sheets("Out(RW)").Select 
02930         Case Else 
02940             GoTo BreakFromHere               ' ストッパー 
02950     End Select 
02960  
02970 ' 書き出しの間は、行を非表示（時間の短縮化のため） 
02980 Rows("3:20000").EntireRow.Hidden = True 
02990  
03000 ' 選択された出力用シートの 下端を探す 
03010 Range("a3").Select 
03020 xxx = Selection.Value 
03030 Do Until xxx = "" 
03040     ActiveCell.Offset(1, 0).Select 
03050     xxx = Selection.Value 
03060 Loop 
03070  
03080 ' 書き出し 
03090 If GoodCounter <> 0 Then 
03100     For i = 1 To Counter 
03110     ColSft = 0 
03120     If GoodSpec(i) = True Then 
03130             Selection.Value = i 
03140                 ActiveCell.Offset(0, 1).Select 
03150                 ColSft = ColSft + 1 

03160             Selection.Value = Flags(i) 
03170                 ActiveCell.Offset(0, 1).Select 
03180                 ColSft = ColSft + 1 
03190             Selection.Value = Judgement(i) 
03200                 ActiveCell.Offset(0, 1).Select 
03210                 ColSft = ColSft + 1 
03220             Selection.Value = DepVar(i) 
03230                 ActiveCell.Offset(0, 1).Select 
03240                 ColSft = ColSft + 1 
03250             Selection.Value = IndepVar(i, 1) 
03260                 ActiveCell.Offset(0, 1).Select 
03270                 ColSft = ColSft + 1 
03280             Selection.Value = IndepVar(i, 2) 
03290                 ActiveCell.Offset(0, 1).Select 
03300                 ColSft = ColSft + 1 
03310             Selection.Value = IndepVar(i, 3) 
03320             If WithWithoutLag = 1 Then 
03330                     ActiveCell.Offset(0, 1).Select 
03340                     ColSft = ColSft + 1 
03350                 Selection.Value = IndepVar(i, 4) 
03360             Else 
03370             End If 
03380             ActiveCell.Offset(1, -ColSft).Select 
03390         Else 
03400     End If 
03410      
03420     Next i 
03430     Else 
03440         Sheets("Base").Select 
03450 End If 
03460  
03470 ' 行の再表示 
03480 Rows("1:20000").EntireRow.Hidden = False 
03490  
03500 ' ------------------------------------ Summary Statistics ------------------------------------ 
03510 ' サマリーシートへの情報書き出し（１／２） 
03520 Sheets("Summary").Select 
03530  
03540 ' 下端を探す 
03550 Range("a1").Select 

03560 xxx = Selection.Value 
03570 Do Until xxx = "" 
03580     ActiveCell.Offset(1, 0).Select 
03590     xxx = Selection.Value 
03600 Loop 
03610  
03620 ' 相手国、財、輸入価格、ラグ項有無 
03630 Selection.Value = PCntry 
03640 ActiveCell.Offset(0, 1).Select 
03650 Selection.Value = ComCode 
03660 ActiveCell.Offset(0, 1).Select 
03670 Selection.Value = RelativePM 
03680 ActiveCell.Offset(0, 1).Select 
03690 Selection.Value = WithWithoutLag 
03700  
03710 ' t 値 Critical Value 
03720 ActiveCell.Offset(0, 1).Select 
03730 Selection.Value = T_Critical + T_Increment 
03740  
03750 ' 定式化番号 
03760 ActiveCell.Offset(0, 1).Select 
03770 Selection.Value = GoodCounter 
03780  
03790 ' Regression シートに移動 
03800 Sheets("Regression").Select 
03810 Next RelativePM ' --- 絶対/相対輸入価格（ループ 4/5：終端) ------------ 
03820 ' -------------------------- Relative PM Loop ----------------------------------------------- 
03830 Next WithWithoutLag ' ----- ラグ項有/無（ループ 3/5：終端) ------------ 
03840 ' --------------------------- With/Wighout Lag Loop ------------------------------------- 
03850 Next ComCode ' --------------- 次の財へ（ループ 2/5：終端) ----------------- 
03860 ' --------------------------- ComCode Loop -------------------------------------------------- 
03870  
03880 BreakFromHere:   ' ======= エラー・緊急時スキップ先 ========= 
03890  
03900 ' 計算終了時刻設定 
03910 Endtime = Time 
03920  
03930 If SingleSpecFlag = True Then ' 国・財条件付きの時は時間表示しない 
03940     Else                               ' 無条件の場合 
03950          ' サマリーシートへの情報書き出し（２／２） 

03960         Sheets("Summary").Select 
03970         ' 下端を探す 
03980         Range("a1").Select 
03990         xxx = Selection.Value 
04000         Do Until xxx = "" 
04010             ActiveCell.Offset(1, 0).Select 
04020             xxx = Selection.Value 
04030         Loop 
04040  
04050         Selection.Value = PCntry 
04060         ActiveCell.Offset(0, 1).Select 
04070         Selection.Value = BeginTime 
04080         ActiveCell.Offset(0, 1).Select 
04090         Selection.Value = Endtime 
04100 End If 
04110  
04120 SkipThisCountry:   ' ===== 相手国が自分自身の場合のスキップ先  ===== 
04130 ‘ Next ii  ' ------- 次の相手国へ（ループ 1/5：終端) ------------ （コメントアウト） 
04140 Next zi '------------------- Selected Country Loop --------------------------------- 
04150  
04160 ' カーソルを先頭に復帰 
04170 Range("a2").Select 
04180  
04190 ThisWorkbook.Save 
04200  
04210 EndofThisMacro:   ' ========== 緊急避難用ジャンプ先 ========== 
04220  
04230 End Sub 
 




